

## Thermocouple model functions for voltage, temperature, derivative and tolerance

D. A. Burkett  
4 mar 04

### Summary

Use these functions to calculate thermocouple temperature, voltage, tolerances and the derivative  $dEMF/dT$  for eleven thermocouple types: J, K, E, T, S, R, N, B, C, D and G. The relations for temperature and EMF come from the *Omega Complete Temperature Measurement Handbook and Encyclopedia*, 21st Century 2nd edition, 2000. I corrected a few errors in the *Handbook* functions and developed functions for thermocouple error and  $dEMF/dT$ . If you need these functions, you what to do with them, so I provide no background information.

All functions use temperatures in °C and thermal EMFs in volts. The reference junction temperature is 0°C. I use the abbreviation *TC* for thermocouple, and *EMF* for the thermal electromotive force (voltage) generated by the junction.

The model functions are corrected for the International Temperature Scale of 1990 (ITS-90). The source data can be found at [www.omega.com](http://www.omega.com). I refer to my TI-89/TI-92 Plus/Voyage 200 tip list, which can be found at [http://www.angelfire.com/realm/ti\\_tiplist/](http://www.angelfire.com/realm/ti_tiplist/).

### Installing the model functions

Install the functions in any folder, unless you use the error function *tc\_err()*. In that case the functions must be installed in folder *thermcpλ*. *tc\_err()* uses matrix *tc\_errm* which also must be installed in folder *thermcpλ*. *tc\_err()* calls *util/casel()*, which is included in this package. *tc\_err()* also calls the corresponding EMF function for the thermocouple type, for example, to find the errors for a type K TC, *tc\_err()* will call *tck\_e()*.

All routines can be archived.

You need only install the functions for the thermocouples you use. For example, if you only use type J TCs, then you can get by with *tcj\_e()*, *tcj\_t()* to find EMF and temperature, and *tcj\_d()* if you are interested in the derivative. The error function *tc\_err()* works properly as long as you do not specify TC types for which the individual functions are not installed.

### Using the model functions

Function names have the general form *tc{type}\_{op}()*, where *{type}* is a single character to specify the TC type, and *{op}* is a single character to specify the function operation. *{type}* may be *j, k, e, t, s, r, n, b, c, d* or *g*. *{op}* may be *e, t* or *d*, to find the EMF, temperature or derivative, respectively. For example:

|                    |   |
|--------------------|---|
| <i>tck_e(100)</i>  | returns the output voltage for a type K TC at 100°C |
| <i>tcj_t(.005)</i> | returns the type J TC temperature at an EMF of 5mV  |
| <i>tcs_d(100)</i>  | returns the type S derivative (in V/°C) at 100°C    |

The functions test the input arguments against range limits. If the argument exceeds the limit, then *undef* is returned. Calling routines can compare the returned result to *undef* and take appropriate action. The range limits for temperature and EMF are shown in Table 1 below.

Most of the model functions consist of two or more functions to cover the entire range. There are discontinuities at the function boundaries, and I have made no effort to splice them smoothly. The discontinuities are usually smaller than the errors of the functions, so this should not be a problem for most uses. If you require a smooth transition from one function to another, you could try the fourth-order splice described in tip [6.56] of the TI-89 tip list. Table 2 below lists the range boundaries and the discontinuities at those boundaries.

### Using the error function

The function *tc\_err()* finds the error limits at a given temperature:

```
tc_err(type_string,temperature_°C,tol_string)
```

where

- type\_string* is a single character string which specifies the TC type, which may be upper- or lower-case, for example, "k" or "K"
- temperature\_°C* is the temperature in °C
- tol\_string* specifies the TC tolerance: "std" for standard wire, any other string for special tolerance wire.

*tc\_err()* returns a list of the form {*temp\_err*, *EMF\_err*} where *temp\_err* is the temperature error, and *EMF\_err* is the equivalent thermoelectric voltage error. For example,

```
tc_err("k",100,"std")
```

find the errors for a standard type K TC at 100°C, and returns {2.2, 91.05E-6}, indicating that the tolerance is ±2.2°C, or ±91.05 µV. If you want to find the tolerances at a given voltage instead of temperature, use the corresponding function to convert voltage to temperature, for example

```
tc_err("j",thermcp1\tcj_t(.027),"special")
```

returns the tolerances for a low-tolerance type J TC at an output of 27mV.

*tc\_err()* can return *undef* or several error messages instead of tolerance list, as shown in this table:

***tc\_err()* Error Messages**

| <b>Result</b>             | <b>Cause</b>  |
|---------------------------|---|
| {undef,undef}             | <i>temperature_°C</i> is undef, or<br><i>temperature_°C</i> exceeds model limits  |
| "tc_err fault, TC type"   | <i>type_string</i> is not a string, or<br><i>type_string</i> is not a single character, or<br><i>type_string</i> is not a valid TC type |
| "tc_err fault, temp"      | <i>temperature_°C</i> is not a number   |
| "tc_err fault, tolerance" | <i>tol_string</i> is not a string   |

*tc\_err()* calls function *util\case()*, and uses matrix *thermcpl\tc\_errm*. *util\case()* is described in the *TI-89/TI-92 Plus/Voyage 200 Tip List* at tip [8.6]. The matrix is defined in a section below.

### Thermocouple temperature ranges and errors

This table specifies the TC ranges and tolerances as they relate to the model functions.

**Table 1**  
**TC ranges and tolerances**

| TC Type | Temperature range °C | EMF range mV         | Standard wire Tolerance                       | Special wire Tolerance                 |
|---------|----------------------|----------------------|---|--|
| J       | -210                 | -8.095               | T < 293.33: 2.2°C                             | T < 275°C: 1.1°C                       |
|         | 1200                 | 69.553               | T > 293.333: 0.75%                            | T > 275°C: 0.4%                        |
| K       | -270                 | -6.458               | T < -110: 2%                                  | T < 275°C: 1.1°C                       |
|         | 1372                 | 54.886               | -110 < T < 293.33: 2.2°C<br>T > 293.33: 0.75% | T > 275°C: 0.4%                        |
| E       | -270                 | -9.835               | T < -170: 1%                                  | T < -250: 0.4%                         |
|         | 1000                 | 76.373               | -170 < T < 340: 1.7°C<br>T > 340: 1.7%        | -250 < T < 250: 1°C<br>T > 250: 0.4%   |
| T       | -270                 | -6.258               | T < -66.67: 1.5%                              | T < -125: 0.4%                         |
|         | 400                  | 20.872               | -66.67 < T < 133.33: 1°C<br>T > 133.33: 0.75% | -125 < T < 125: 0.5°C<br>T > 125: 0.4% |
| S       | -50                  | -2.3555              | T < 600: 1.5°C                                | T < 600: 0.6°C                         |
|         | 1768.1               | 18.6935              | T > 600: 0.25%                                | T > 600: 0.1%                          |
| R       | -50                  | -0.22647             | T < 600: 1.5°C                                | T < 600: 0.6°C                         |
|         | 1768.1               | 21.1027              | T > 600: 0.25%                                | T > 600: 0.1%                          |
| N       | -270                 | -4.345               | T < -110: 2%                                  | T < 275: 1.1°C                         |
|         | 1300                 | 47.513               | -110 < T < 293.33: 2.2°C<br>T > 293.33: 0.75% | T > 275: 0.4%                          |
| B       | 0                    | -0.0257938           | T < 800°C: not specified                      | Not available                          |
|         | 1820                 | 13.82028<br>(Note 1) | T > 800: 0.5°C                                |  |
| C       | -17.778              | -0.0234471           | T < 425: 4.5°C                                | Not available                          |
|         | 2315.556             | 37.06598             | T > 425: 1%                                   |  |
| D       | -17.778              | -0.163494            | Not specified                                 | Not available                          |
|         | 2320                 | 39.5472              |   |  |
| G       | -17.778              | -0.0158              | Not specified                                 | Not available                          |
|         | 2315.556             | 38.5644              |   |  |


Note (1): Type B voltage as a function of temperature is non-monotonic, with a minimum of -2.584972uV at a temperature of 21.020262°C. In consequence, function *tc\_b\_f()* has a lower range limit of about -2.579382uV, at 22°C.

### Model function temperature errors, breakpoints and discontinuities

Table 2 below summarizes the error limits for the model functions  $T = f(\text{EMF})$ . The model functions for  $\text{EMF} = f(T)$  are considered 'exact', but the inverse functions are approximations. The error limits in the table are conservative, and the errors over some temperature ranges can be much less. The error for a particular temperature range can be found by plotting the error expression

$$t_{cy\_t}(t_{cy\_e}(x)) - x$$

where  $y$  is the thermocouple type. This plot shows the error for a type K thermocouple from -200 to 1300 °C. The  $y$ -axis range is -0.05 to 0.04 °C.



The table also summarizes the model function range breakpoints and the discontinuities at those breakpoints. In most cases the temperature discontinuities are on the order of the temperature errors, which isn't too impressive, but the functions are still useful. It does make the point, though, that these functions are more appropriate for general engineering use than precision thermometry.

**Table 2**  
**TC temperature errors, breakpoints and discontinuities**

| TC Type | T=f(EMF) error (°C) | T = f(EMF) breakpoints (°C) | Discontinuity at EMF breakpoint (V) | EMF = f(T) breakpoints (mV) | Discontinuity at temperature breakpoint (°C) |
|---------|---------------------|-----------------------------|-------------------------------------|-----------------------------|--|
| J       | ±0.05               | 760                         | 7.49 E-11                           | 0<br>42.919                 | 0<br>0.0675                                  |
| K       | ±0.06               | 0                           | 1.974 E-12                          | -5.891<br>0<br>20.644       | 0.0405<br>0<br>0.0331                        |
| E       | ±0.03               | 0                           | 0                                   | -8.825<br>0                 | 0.0218<br>0                                  |
| T       | ±0.04               | 0                           | 0                                   | -5.603<br>0                 | 0.0354<br>0                                  |
| S       | ±0.02               | 1064.18<br>1664.5           | 9.942 E-12<br>3.354 E-9             | 1.874<br>11.950<br>17.536   | 0.001883<br>0.00967<br>0.00130               |
| R       | ±0.02               | 1064.18<br>1664.5           | 1.7 E-14<br>1.72 E-12               | 1.923<br>13.228<br>19.739   | 0.00875<br>0.00456<br>0.000833               |
| N       | ±0.06               | 0                           | 0                                   | -3.998<br>0<br>20.613       | 0.0352<br>0<br>0.0118                        |
| B       | ±0.03               | 630.615                     | 4.402 E-7                           | 0.2913<br>2.4306            | 0.0255<br>0.0264                             |
| C       | ±0.013              | None                        | n/a                                 | 14.02                       | 0.00204                                      |
| D       | ±0.03               | 783                         | 4.377 E-8                           | 2.887<br>3.8277             | 0.0127<br>0.00744                            |
| G       | ±0.04               | None                        | n/a                                 | 0.03088<br>3.7202           | 0.0113<br>0.000373                           |

### Comments on the model equations

I have implemented the functions for TC EMF as given by the *Omega Handbook*. The *Handbook* also gives functions for TC temperature as a function of EMF, but for some TC types the functions do not cover the same temperature range as the EMF functions. For those TCs I found approximating functions to provide guesses for *nSolve()* to solve for the temperature. This means that solutions will be slower for those TC types, at low temperatures.

The derivative functions *tcx\_d()* use the 'exact' EMF = f(T) models, so the derivatives are dEMF/dT V/°C. To find dT/dEMF just use 1/(dEMF/dT). The derivative error limits can be estimated with

$$\left(\frac{dEMF}{dT}\right)_{\min} = \frac{d}{dT}f(T - T_e) \qquad \left(\frac{dEMF}{dT}\right)_{\max} = \frac{d}{dT}f(T + T_e)$$

where  $T_e$  is the temperature error at T. As an example, find the derivative limits for a standard type K TC at 500 °C. First find the temperature error with

```
tc_err("k",500,"std")[1]
```

which returns 3.75 so the temperature error is ±3.75 °C. Then find the derivative limits with

```
tck_d(500-3.75)    returns 4.2622 38 E-5
tck_d(500+3.75)    returns 4.2633 53 E-5
```

The nominal derivative at 500°C is 4.2628 33 E-5, so the worst-case limit is ±5.95 E-9 V/°C.

The equations for the minimum and maximum derivatives at a temperature arise from the fact that the nominal model function EMF = f(T) is bound above and below by functions  $f_1(T)$  and  $f_2(T)$ , where  $f_1(T) = f(T) + E_e$  and  $f_2(T) = f(T) - E_e$ . We want  $f_1'(T)$  and  $f_2'(T)$ .  $E_e$  is the EMF error which we assume is the same for  $f_1$  and  $f_2$ . Now,  $E_e = f(T+T_e) - f(T)$ , where  $T_e$  is the temperature error. So  $f_1(T) = f(T+T_e)$ , and  $f_1'(T) = f'(T+T_e)$ .

The temperature tolerance error is found from applying the tolerance in Table 1. The equivalent EMF errors are found by calculating the EMFs at the temperature tolerance extrema and returning the largest EMF error. For example, suppose that the temperature tolerance at temperature T is  $T_d$ , then the temperature limits are  $T_1 = T - T_d$  and  $T_2 = T + T_d$ . If the EMFs corresponding to T,  $T_1$  and  $T_2$  are E, E1 and E2, respectively, then the EMF errors are  $|E - E1|$  and  $|E - E2|$ .

Some TC types (S, R, B and C) exhibit a peak in the EMF error near the maximum temperature. While it is physically unlikely that the EMF reaches a maximum, I have not modified the calculation prevent it. The table below shows the errors and associated temperatures.

To determine the conditions for an error peak, let EMF = f(T) and define the temperature tolerance as a multiplier  $n$  such that  $T_e = nT$  so that  $T_e$  is the maximum-tolerance temperature at T. For example, if the tolerance specification is 2%, then  $n = 1.02$ . Define the EMF error as

$$EMF_{err} = e(T) = f(T) - f(T_e)$$

Take the derivative and set to zero to find the maximum:

$$\frac{d}{dT}e(T) = \frac{d}{dT}f(T) - \frac{d}{dT}f(nT) = 0 \qquad \text{or} \qquad \frac{d}{dT}f(T) = \frac{d}{dT}f(nT)$$

If this condition is met for some T within the operating temperature range, then there will be an error peak. The analysis is somewhat different for the type B TC, as the error is specified as a flat 0.5°C, instead of a percentage tolerance, but the basic idea is the same.

**Peak EMF tolerance errors and temperatures**

| TC Type     | Max T       | EMF error at max T | EMF error peak T | EMF error at peak T |
|-------------|-------------|--------------------|------------------|---------------------|
| S, standard | 1768.1 °C   | 45.8033 uV         | 1685.184 °C      | 48.8775 uV          |
| S, special  | 1768.1 °C   | 18.2671 uV         | 1685.184 °C      | 19.5206 uV          |
| R, standard | 1768.1 °C   | 54.4105 uV         | 1689.912 °C      | 57.3355 uV          |
| R, special  | 1768.1 °C   | 21.7072 uV         | 1688.645 °C      | 22.9172 uV          |
| B           | 1820 °C     | 5.7098 uV          | 1634.6054 °C     | 5.8520 uV           |
| C           | 2315.556 °C | 215.1205 uV        | 1900.0924 °C     | 246.6944 uV         |

*Matrix tc\_errm description*

tc\_errm is an 11-row, 8-column matrix which holds the tolerances and temperature limits for all the TC types. tc\_err() uses this matrix in place of a lot of hard-coded constants to simplify the code. Not all elements are relevant for all TC types.

**tc\_err matrix contents**

| TC Type | EAsstd T>0 | ERstd T>0 | EAsstd T<0 | ERstd T<0 | EAspc | ERspc | Tmin   | Tmax     |
|---------|------------|-----------|------------|-----------|-------|-------|--------|----------|
| J       | 2.2        | 0.0075    | 2.2        | 0.0075    | 1.1   | 0.004 | -210   | 1,200    |
| K       | 2.2        | 0.0075    | 2.2        | 0.02      | 1.1   | 0.004 | -270   | 1,372    |
| E       | 1.7        | 0.005     | 1.7        | 0.01      | 1     | 0.004 | -270   | 1,000    |
| T       | 1          | 0.0075    | 1          | 0.015     | 0.5   | 0.004 | -270   | 400      |
| S       | 1.5        | 0.0025    | 1.5        | 0.0025    | 0.6   | 0.001 | -50    | 1,768.1  |
| R       | 1.5        | 0.0025    | 1.5        | 0.0025    | 0.6   | 0.001 | -50    | 1,768.1  |
| N       | 2.2        | 0.0075    | 2.2        | 0.02      | 1.1   | 0.004 | -270   | 1,300    |
| B       | 0          | 0         | 0          | 0         | 0     | 0     | 0      | 1,820    |
| C       | 0          | 0         | 0          | 0         | 0     | 0     | -17.78 | 2,315.56 |
| D       | 0          | 0         | 0          | 0         | 0     | 0     | -17.78 | 2,320    |
| G       | 0          | 0         | 0          | 0         | 0     | 0     | -17.78 | 2,315.56 |

The column descriptions are:

- EAsstd T>0      Absolute error in °C for standard wire, T > 0°C
- ERstd T>0      Relative error for standard wire, T > 0°C
  
- EAsstd T<0      Absolute error in °C for standard wire, T < 0°C
- ERstd T<0      Relative error for standard wire, T < 0°C
  
- EAspc            Absolute error in °C for special limits-of-error wire
- ERspc            Relative error for special limits-of-error wire

|      |                           |
|------|---------------------------|
| Tmin | Minimum model temperature |
| Tmax | Maximum model temperature |


### Some examples

Example 1: Compare the linearity of a type J to type K thermocouple over the range 0 to 1000°C.

A perfectly linear function would have a constant derivative, so we plot the derivatives over the desired temperature range. From the command line or in the Y= editor, set

```
tck_d(x)→y1(x)
tcj_d(x)→y2(x)
```

Set *xmin* to 0, *xmax* to 1000, then ZoomFit to get this plot:



The upper trace is the type J TC, the lower trace is the type K. The type J is more sensitive (more mV output for each °C), but the type K is closer to linear.

Example 2: A temperature measurement system is subject to noise with a peak amplitude of 30 uV at the type E TC input connector. Find the equivalent worst-case temperature error over a 200 to 400° measurement range.

Again we use the derivative function to find the sensitivity. *tct\_d(200)* returns 53.15uV/°C, *tct\_d(400)* returns 61.8 uV/°C, so the worst case is at 400°C. Then, (30 uV)/(61.8 uV/°C) gives a temperature uncertainty of about 0.5°C.

Example 3: Find a 2nd-order approximation  $T=f(\text{EMF})$  for a type K transfer function over a temperature range of -10 to 100°C. Find the worst-case approximation error.

Try fitting 20 points:

```
seq(x,x,-10,100,(100-^-10)/19)→tlist
seq(tck_e(tlist[k]),k,1,20)→elist
quadreg elist,tlist
regeq(elist)-tlist→resid
```

The minimum and maximum values of the fit residuals list *resid* are -0.161 and 0.158°C, so the worst-case approximation error is 0.161°C. The fit function coefficients are in the system variable *regcoef*.

## Source code - `tc_err()`

Annotated source code for `tc_err()` is shown below. The code is completely straightforward with the possible exception of building the function calls for the EMF routines, to calculate the EMF error.

```
tc_err( $\tau$ type, $\tau$ c, $\tau$ tol)
Func
@("TC type",T°C,"std" or "special")
@Return {temp_err°C,emf_err volts}
@Calls util\case1(), uses thermcpl\tc_errm
@29mar04/dburkett@infinet.com

local  $\tau$ err, $\epsilon$ mferr, $\tau$ i, $\epsilon$ a, $\epsilon$ r, $\epsilon$ 1, $\epsilon$ 2, $\epsilon$ t, $\tau$ min, $\tau$ max

@  $\tau$ err      temperature error
@  $\epsilon$ mferr   EMF error
@  $\tau$ i       index into tc_errm matrix
@  $\epsilon$ a      absolute temperature error
@  $\epsilon$ r      relative temperature error
@  $\epsilon$ 1      EMF at  $\tau$ min
@  $\epsilon$ 2      EMF at  $\tau$ max
@  $\epsilon$ t      EMF at  $\tau$ c
@  $\tau$ min     Temperature minimum error bound
@  $\tau$ max     Temperature maximum error bound

@ Validate input arguments. Test temperature to undef, as tcx_t() functions may return
undef. In this case, return {undef,undef} to prevent faults when plotting, or to
propogate undef through subsequent calculations. Otherwise, ensure that TC type is a
single-character string, temperature is a number, and tolerance type is a string.

if  $\tau$ c=undef : return {undef,undef}
if gettype( $\tau$ type)≠"STR"
  return "tc_err fault, TC type"
if dim( $\tau$ type)≠1
  return "tc_err fault, TC type"
if gettype( $\tau$ c)≠"NUM"
  return "tc_err fault, temp"
if gettype( $\tau$ tol)≠"STR"
  return "tc_err fault, tolerance"

@ Convert tolerance and TC type to lower-case for subsequent comparisons. Convert the TC
type to index  $\tau$ i, and return error string if TC type not valid.

util\case1( $\tau$ tol)→ $\tau$ tol
util\case1( $\tau$ type)→ $\tau$ type
instring("jketsrnbcdg", $\tau$ type)→ $\tau$ i
if  $\tau$ i=0
  return "tc_err fault, TC type"

@ Find the minimum and maximum temperatures for the TC type and return {undef,undef} if
@ temperature exceeds limits. Returning undef prevents faults when plotting errors.
thermcpl\tc_errm[ $\tau$ i,7]→ $\tau$ min
thermcpl\tc_errm[ $\tau$ i,8]→ $\tau$ max

if  $\tau$ c< $\tau$ min or  $\tau$ c> $\tau$ max
  return {undef,undef}

@ Types J, K, E, T, S, R, and N can all be handled in the same way. Find the relative and
absolute error specifications  $\epsilon$ r and  $\epsilon$ a from the tc_errm matrix based on the wire
tolerance and temperature. Temperature error  $\tau$ err is the larger of the absolute or
relative error. Find the EMF limits  $\epsilon$ 1 and  $\epsilon$ 2 from the temperature limits, then the
maximum EMF error.

if  $\tau$ i≤7 then
  if  $\tau$ tol="std" then
    if  $\tau$ i<0 then
      thermcpl\tc_errm[ $\tau$ i,3]→ $\epsilon$ a
```

```

        thermcpl\|tc_errm[\tau_i,4]→εr
    else
        thermcpl\|tc_errm[\tau_i,1]→εa
        thermcpl\|tc_errm[\tau_i,2]→εr
    endif
else
    thermcpl\|tc_errm[\tau_i,5]→εa
    thermcpl\|tc_errm[\tau_i,6]→εr
endif

max({εa,εr*τc})→τerr

max({τc-τerr,τmin})→τmin
min({τc+τerr,τmax})→τmax

expr("thermcpl\|tc"&type&"_e(τmin)")→ε1
expr("thermcpl\|tc"&type&"_e(τmax)")→ε2
expr("thermcpl\|tc"&type&"_e(τc)")→εt
max(abs({ε1-εt,ε2-εt}))→εmferr

return {τerr,εmferr}

endif

© Type B is a special case since the temperature error is a constant 0.5C above 800C.
if type="b" then
    if τc<800
        return {undef,undef}

    .5→τerr

    max({τc-τerr,τmin})→τmin
    min({τc+τerr,τmax})→τmax

    thermcpl\|tcb_e(τmin)→ε1
    thermcpl\|tcb_e(τmax)→ε2
    thermcpl\|tcb_e(τc)→εt

    max(abs({ε1-εt,ε2-εt}))→εmferr

    return {τerr,εmferr}

endif

© Type C is a special case because of the error boundary point
if type="c" then

    when(τc<425,4.5,.01*τc)→τerr

    max({τc-τerr,τmin})→τmin
    min({τc+τerr,τmax})→τmax

    thermcpl\|tcc_e(τmin)→ε1
    thermcpl\|tcc_e(τmax)→ε2
    thermcpl\|tcc_e(τc)→εt

    max(abs({ε1-εt,ε2-εt}))→εmferr

    return {τerr,εmferr}

endif

© Error not specified for types D and G
if type="d" or type="g"
    return {undef,undef}

EndFunc

```

### Source code - a typical set of model functions

I won't list the source code for all the model functions, as that would be repetitive with little benefit. However, the three functions for the type K TC show the typical structure of all the functions.

First, the function to find the TC EMF, given temperature, consists of no more than a temperature range limit test and a test for an *undef* input argument, followed by evaluation the appropriate polynomial to find the EMF. As the Omega reference functions usually return the EMF in uV, I scale the result to return volts.

```
tck_e(τc)
Func
@(temp °C) Type K TC EMF
@13feb04/dburkett@infinet.com

if τc=undef : return undef
if τc<-270 or τc>1372 : return undef

when(τc<0,polyeval({-1.6322697486E-20,-1.9889266878E-17,-1.0451609365E-14,-3.1088872894E-12,-5.7410327428E-10,-6.7509059173E-8,-4.9904828777E-6,-3.2858906784E-4,2.3622373598E-2,39.450128025,0},τc)*10^-6,(polyeval({-1.2104721275E-23,9.7151147152E-20,-3.2020720003E-16,5.6075059059E-13,-5.6072844889E-10,3.1840945719E-7,-9.9457592874E-5,1.8558770032E-2,38.921204975,-17.600413686},τc)+118.5976*e^(-1.183432E-4*(τc-126.9686)^2))*10^-6)

EndFunc
```

Some of the functions to find temperature given EMF are more involved because the Omega reference does not give model functions at low EMFs. I get around this by using *nSolve()* with an approximating function, as shown below.

```
tck_t(εmf)
Func
@(EMF V) Type K TC temperature °C
@13feb04/dburkett@infinet.com

local τemf,εmfuv

© Define a function to return EMF=f(t), which is just taken from tck_e(), for the appropriate temperature range.

define τemf(τ1)=func
polyeval({-1.6322697486E-20,-1.9889266878E-17,-1.0451609365E-14,-3.1088872894E-12,-5.7410327428E-10,-6.7509059173E-8,-4.9904828777E-6,-3.2858906784E-4,2.3622373598E-2,39.450128025,0},τ1)*10^-6
endfunc

© Test the EMF for undef, return same if so. Convert EMV to uV. Return undef if EMF is out of range.

if εmf=undef : return undef
εmf*10^6→εmfuv
if εmfuv<-6458 or εmfuv>54886
return undef

© Omega reference does not give corresponding function for T = f(EMF) for EMF < 5891 uV, so use nSolve() to find T. A rational polynomial estimating function gives a close guess for nSolve(), to speed execution time and ensure the correct root is returned.

if εmfuv<-5891 then
return
nSolve(τemf(τ)=εmf,τ=polyeval({-10589.5276,-69.0356203},εmf)/polyeval({16980.82020,264.1352642,1},εmf))
© For remaining EMF ranges just implement polynomials given by Omega reference.
```

```

elseif emfuv<0 then
return
polyeval ({-5.1920577E-28,-1.0450598E-23,-8.6632643E-20,-3.7342377E-16,-8.977354E-13,-1.08
33638E-9,-1.1662878E-6,2.5173462E-2,0},emfuv)

elseif emfuv<20644 then
return
polyeval ({-1.052755E-35,1.057734E-30,-4.413030E-26,9.804036E-22,-1.228034E-17,8.31527E-14
,-2.503131E-10,7.860106E-8,2.508355E-2,0},emfuv)

else
return
polyeval ({-3.110810E-26,8.802193E-21,-9.650715E-16,5.464731E-11,-1.646031E-6,4.830222E-2,
-131.8058},emfuv)

endif

EndFunc

```

The derivative functions are straightforward conversions of the EMF = f(T) functions. The derivatives of the estimating polynomials are coded, using the same range boundaries as the original f(T). Again, the functions return *undef* for an *undef* input, and return *undef* if the input temperature is out of range.

```

tck_d(tc)
Func
@(temp °C) Type K TC dEMF/dT
©14feb04/dburkett@infinet.com

if tc=undef : return undef
if tc<-270 or tc>1372
return undef

10^-6*when(tc<0,polyeval ({-1.6322697486E-19,-1.79003401902E-16,-8.361287492E-14,-2.176221
10258E-11,-3.4446196456801E-9,-3.37545295865E-7,-1.9961931510801E-5,-9.8576720352E-4,.047
244747196,39.450128025},tc),polyeval ({-1.08942491475E-22,7.77209177216E-19,-2.24145040021
E-15,3.36450354354E-12,-2.80364224445E-9,1.27363782876E-6,-.000298372778622,.037117540064
,38.921204975},tc)-.028070438992704*(tc-126.9686)*(.99988166380228)^((tc-126.9686)^2))

EndFunc

```

### Source code - *casel()* utility

*casel()* is used by *tc\_err()* to convert input arguments to lower-case for subsequent testing. It must be installed in the *lutil* folder.

```

casel(s)
Func
© (string) convert to lower case
© 7may02/dburkett@infinet.com

local ä,ï,ÿ

""->ÿ

for ä,1,dim(s)
ord(mid(s,ä,1))->ï
ÿ&char(when(ï≥65 and ï≤90 or ï≥192 and ï≤214 or ï≥216 and ï≤223,ï+32,ï))->ÿ
endfor

return ÿ

EndFunc

```